An Introduction to *Maple*

Graphing Functions $y = f(x)$ by Bob Bradshaw, Ohlone College, Fremont, CA

To graph a function like $y = x^2$, use plot(*function*, x=*start..end)*

```
> plot(x^2,x=-4..4);
```

You can also specify the *y* values.

```
plot(x^2,x=-4..4,y=-10..25);
```

Instead of enterring the functions within the plot command, you can define the functions first.

```
> y1:=x^3;
  y2:=4-x^2;
  plot([y1,y2],x=-4..4);
```

By clicking on the graph, you can use the menus and the icons to alter the appearance of the graph.

The appearance of the graph can also be changed within the plot statement by using a series of options.

```
> plot([y1,y2],x=-4..4,color=[green,blue]);
> plot([y1,y2],x=-4..4,color=[green,blue],thickness=[3,6]);
> plot([y1,y2],x=-4..4,color=[green,blue],style=point,
  symbol=[box,cross]);
> plot([y1,y2],x=-4..4,color=[green,blue],linestyle=[2,3],
  thickness=2);
> plot(y1,x=-4..4,labels=[Time,
  Temperature],labelfont=[TIMES,BOLD,18]);
> plot(y1,x=-4..4,labels=[Time,
  Temperature],labelfont=[TIMES,BOLD,14],title="My
  Graph",titlefont=[TIMES,BOLD,18]);
```

The plotting routines are not perfect. This is supposed to be the graph of a circle.

```
> y3:=sqrt(9-x^2);
  y4:=-y3;
  plot([y3,y4],x=-10..10);
```

Using the option *scaling=constrained* makes the graph look like a circle rather than an ellipse.

```
> plot([y3,y4],x=-10..10,scaling=constrained);
```

Increasing the number of points attempts to join the ends of the semicircles.

```
> plot([y3,y4],x=-10..10,numpoints=5000,scaling=constrained);
```

Maple can create and graph piecewise continuous functions

```
> y6:= piecewise(x<2,x,x>=2 and x<5,9, x>=5,10-x);
> plot(y6,x=-5..10,thickness=2);
```

In order to produce a graph that shows the discontinuity, use the option "discont=true"

```
 plot(y6,x=-5..10,discont=true,thickness=2);
```

Maple can also create a graph of series of points. First, create a list of *x* and *y* values

```
> xvalues:=[1,2,3,4,5];
> yvalues:=[3,2,6,7,1];
```

Next, turn the two lists into a list of points.

```
> points:=[[xvalues[n],yvalues[n]]$n=1..5];
```

Finally, graph the points with the plot command.

```
> plot(points,x=0..8,style=point,symbol=circle,view=[0..8,0..8]);
```

You can combine the graph of the points with any other graph.

```
plot([points,3+sin(x)],x=0..8,style=[point,line],symbol=circle,
color=[green,red],thickness=2);
```

There are additional plotting commands that are available in *Maple*. To access many of these, you must first load the plotting subroutine using the following command.

```
> with(plots);
```

One of the commands in the plotting package is "display." This lets you create and store a series of graphs and then display them with a later statement. To do this, simply give the standard plot statements names like p1, and p2. Be sure to end the statements with colons rather than semicolons. Using a colon will hide the output. If you want to see what happens, try putting a semicolon at the end of the next statement instead of a colon.)

```
> p1:=plot(x^2,x=-2..2,color=green):
  p2:=plot(3*cos(x),x=-4..4,color=red,thickness=3):
```

Finally, create the graph by using the "display" command.

```
> display(p1,p2);
```

You Try it.

1. Graph the following equation. Adjust the x and y values to create an appropriate viewing window.
$y = x^4 - 58\,x^3 + 6\,x^2 - 7\,x + 1$

2. Change the above graph so that the axes are in a large font. Also create a title for the graph.

3. Create a single graph containing the graphs of the following functions on the interval -2*Pi < x < 2*Pi using three different colors and line thicknesses.
$y = \sin(2x)$, $y = 2\sin(2x)$, and $y = 3\sin(2x)$.

4. Create a single graph containing the graphs of the following functions using three different symbols. The graphs should consist of points, not a smooth curve.
$y = \ln(x)$, $y = \ln(2x)$, and $y = \ln(3x)$.

5. Graph the following discontinuous function. Be sure the graph is discontinuous.

$$y := \begin{cases} -5 & x < 2 \\ x^2 & 2 \le x \end{cases}$$